
Subject: Re: Running this MATLAB code in IDL

Posted by [Michael Galloy](#) on Mon, 28 Apr 2014 01:20:16 GMT

[View Forum Message](#) <> [Reply to Message](#)

On 4/27/14, 7:00 pm, mikejohnryan08@gmail.com wrote:

> Thanks for the responses! I agree that it would not be a good idea
> to attempt using the MATLAB code in any way towards thesis research,
> and that it would be better to write my own simple IDL function to do
> this.
>
> Mike, thanks for pointing me towards your code library. The
> "combinelImages" method from your anaglyph class seems to be EXACTLY
> what I am trying to implement in IDL, and seems like it would be a
> very basic function; since as you said, I already have the left and
> right images. But since I am completely new to IDL, I can't even
> begin to think about how that method could be turned into a
> standalone function. It should be very basic, right?

Methods have access to the instance variables of the object and can call other methods of the object using the "self" variable, so watch out for them. In the `_combinelImages` example, there was only the `self.color` variable, so how about something like:

```
function mg_create_anaglyph, leftImage, rightImage, color=color  
  compile_opt strictarr
```

```
  ; define combined_image to the correct size  
  combinedImage = leftImage * 0B  
  dims = size(leftImage, /dimensions)
```

```
  if (keyword_set(color)) then begin  
    combinedImage[0, *, *] = leftImage[0, *, *]  
    combinedImage[1, *, *] = rightImage[1, *, *]  
    combinedImage[2, *, *] = rightImage[2, *, *]  
  endif else begin  
    _leftImage = byte(total(fix(leftImage), 1) / 3)  
    _rightRight = byte(total(fix(rightImage), 1) / 3)
```

```
    combinedImage[0, 0, 0] = Reform(_leftImage, 1, dims[1], dims[2])  
    combinedImage[1, 0, 0] = Reform(_rightRight, 1, dims[1], dims[2])  
    combinedImage[2, 0, 0] = Reform(_rightRight, 1, dims[1], dims[2])  
  endelse
```

```
  return, combinedImage  
end
```

Mike

--

Michael Galloy
www.michaelgalloy.com
Modern IDL: A Guide to IDL Programming (<http://modernidl.idldev.com>)
Research Mathematician
Tech-X Corporation
